

Simplified Stack Machine Assembler Manual

(\$Revision: 1.27 \$)

Gary T. Leavens
Leavens@ucf.edu

August 28, 2024

Abstract

This document defines the assembly language for the Simplified Stack Machine VM, which is used in the Systems Software class (COP 3402) at UCF. It also defines the interface of the assembler and disassembler.

1 Overview

The assembler for the Simplified Stack Machine (SSM) is simple and has no macro facilities. However, it does let one assemble the instructions that make up a program, resolving symbolic names for jump targets, and it can define the starting address of a program and the program's (static) data section. This assembly language adopts some conventions from the MIPS processor's assembly language [1].

The assembler assumes that the SSM has 32-bit (4-byte) words and is word-addressable.

1.1 Inputs and Outputs

1.1.1 Assembler

The assembler is passed a single file name as its only command line argument; this file should be the name of a (readable) assembler program; its output (sent to standard output) is a binary object file.

For example, if the program is contained in the file `myProg.ssm`, assuming that the assembler's executable is named `asm` (and both these files are in the current directory), then the assembler can be invoked as follows in the Unix shell to produce a binary object file on standard output.

```
./asm myProg.ssm
```

Thus, to put the assembled version of `myProg.ssm` into the file `myProg.bof`, one would use a Unix command that redirects the output of the assembler into `myProg.bof`, as follows.

```
./asm myProg.ssm > myProg.bof
```

In addition to the binary object file name, the assembler can be given one command line argument. These options are used for debugging the assembler's code: `-l` to print the lexical tokens read, `-u` to unparse the input's abstract syntax tree (after constructing it), and `-s` to print the symbol table for the file. A binary object file name must always be given after using an option. Only one option can be used for any given invocation of the assembler, and when the `-l` or `-u` options are used, no binary object file is produced.

1.1.2 Disassembler

The disassembler is the opposite of the assembler. It is passed a single file name, but that file names a (readable) binary object file, and it produces, on standard output, an assembly language source program.

For example, if the binary object file is found in `prog.bof`, assuming that the disassembler is named `disasm`, (and both these files are in the current directory), then the disassembler can be invoked as follows in the Unix shell to produce (on standard output) an assembly language program that would compile to `prog.bof`.

```
./disasm prog.bof
```

The assembly language program can be redirected into the file `prog.ssm` as follows.

```
./disasm prog.bof > prog.ssm
```

1.2 Error Outputs

All error messages (e.g., for file permission errors or syntax errors) are sent to standard error output (`stderr`).

1.3 Exit Codes

When the either program halts normally, it exits with a zero error code (which indicates success on Unix). However, when `asm` or `disasm` encounters an error, it halts and exits with a non-zero exit code (which indicates failure on Unix).

2 Assembly Language Syntax

2.1 Lexical Grammar

Tokens in the assembler are described by the (regular) grammar of Figure 1. Note that line endings are significant in the context-free grammar of the assembler, as each instruction must be specified on a single line. Lines may be ended either by a newline character (`<newline>` in Figure 1) or by a combination of a carriage-return (`<cr>`) followed by a newline. Comments (`<comment>`) start with a `<comment-start>` character (i.e., `#`) and continue to the end of a line. White space is needed to separate tokens, but is otherwise ignored.

The lexical grammar (in Figure 1) uses a `terminal font` for terminal symbols. Note that an underbar (`_`) and all ASCII letters (a-z and A-Z) are included in the production for `<letter>`. Curly brackets, such as `{x}`, mean an arbitrary number of (i.e., 0 or more) repetitions of `x`. Note that curly braces are not terminal symbols in the grammar. Some character classes are described in English, these are described in a Roman font between double quotation marks (“ and ”). Note that all characters matched by the non-terminal `<ignored>` are ignored by the lexer. However, the characters that are part of an `<eol>` token (i.e., carriage returns and newlines) are not ignored immediately following a pound-sign (which starts a comment) `<reserved-opcode>` or `<reserved-data-size>`, although they are ignored in all other contexts.

2.2 Context-Free Grammar

The syntax of the SSM’s assembly language is defined by the (context-free) grammar in Figure 2 and Figure 3. The grammar uses a `typewriter font` for terminal symbols.

$\langle \text{section-mark} \rangle ::= .\text{text} \mid .\text{data} \mid .\text{end}$
 $\langle \text{reserved-opcode} \rangle ::= \text{NOP} \mid \text{ADD} \mid \text{SUB} \mid \text{CPW} \mid \text{AND} \mid \text{BOR}$
 $\quad \mid \text{NOR} \mid \text{XOR} \mid \text{LWR} \mid \text{SWR} \mid \text{SCA} \mid \text{LWI} \mid \text{NEG} \mid \text{LIT} \mid \text{ARI} \mid \text{SRI}$
 $\quad \mid \text{MUL} \mid \text{DIV} \mid \text{CFHI} \mid \text{CFLO} \mid \text{SLL} \mid \text{SRL} \mid \text{JMP} \mid \text{CSI} \mid \text{JREL}$
 $\quad \mid \text{ADDI} \mid \text{ANDI} \mid \text{BORI} \mid \text{NORI} \mid \text{XORI} \mid \text{BEQ} \mid \text{BGEZ} \mid \text{BGTZ} \mid \text{BLEZ}$
 $\quad \mid \text{BLTZ} \mid \text{BNE} \mid \text{JMPA} \mid \text{CALL} \mid \text{RTN} \mid \text{EXIT} \mid \text{PSTR} \mid \text{PCH}$
 $\quad \mid \text{RCH} \mid \text{STRA} \mid \text{NOTR}$
 $\langle \text{reserved-data-size} \rangle ::= \text{WORD} \mid \text{CHAR} \mid \text{STRING}$
 $\langle \text{ident} \rangle ::= \langle \text{letter} \rangle \{ \langle \text{letter-or-digit} \rangle \}$ “but not a $\langle \text{reserved-opcode} \rangle$ or $\langle \text{reserved-data-size} \rangle$ ”
 $\langle \text{letter} \rangle ::= _ \mid \text{a} \mid \text{b} \mid \dots \mid \text{y} \mid \text{z} \mid \text{A} \mid \text{B} \mid \dots \mid \text{Y} \mid \text{Z}$
 $\langle \text{letter-or-digit} \rangle ::= \langle \text{letter} \rangle \mid \langle \text{dec-digit} \rangle$

 $\langle \text{unsigned-number} \rangle ::= \langle \text{dec-digit} \rangle \{ \langle \text{dec-digit} \rangle \}$
 $\quad \mid 0\text{x} \langle \text{hex-digit} \rangle \{ \langle \text{hex-digit} \rangle \}$
 $\langle \text{dec-digit} \rangle ::= 0 \mid 1 \mid 2 \mid 3 \mid 4 \mid 5 \mid 6 \mid 7 \mid 8 \mid 9$
 $\langle \text{hex-digit} \rangle ::= \langle \text{dec-digit} \rangle \mid \text{a} \mid \text{A} \mid \text{b} \mid \text{B} \mid \text{c} \mid \text{C} \mid \text{d} \mid \text{D} \mid \text{e} \mid \text{E} \mid \text{f} \mid \text{F}$
 $\langle \text{oct-digit} \rangle ::= 0 \mid 1 \mid 2 \mid 3 \mid 4 \mid 5 \mid 6 \mid 7$

 $\langle \text{reg} \rangle ::= \$ \langle \text{oct-digit} \rangle \mid \$\text{gp} \mid \$\text{sp} \mid \$\text{fp} \mid \$\text{r3} \mid \$\text{r4} \mid \$\text{r5} \mid \$\text{r6} \mid \$\text{ra}$

 $\langle \text{char-literal} \rangle ::= ' \langle \text{char-elem} \rangle '$
 $\langle \text{char-elem} \rangle ::= \langle \text{char} \rangle$ “but not an unescaped ' character”
 $\langle \text{char} \rangle ::= \langle \text{c-escape-seq} \rangle \mid \langle \text{printable-ASCII-char} \rangle$
 $\langle \text{c-escape-seq} \rangle ::= \langle \text{backslash} \rangle \langle \text{escape-code} \rangle$
 $\langle \text{backslash} \rangle ::= \text{“A backslash character (ASCII 92)”}$
 $\langle \text{escape-code} \rangle ::= \text{n} \mid \text{r} \mid \text{f} \mid \text{t} \mid \text{v} \mid \langle \text{backslash} \rangle \mid ' \mid " \mid 0 \mid \text{a} \mid \text{b}$
 $\quad \mid \text{x} \langle \text{hex-digit} \rangle \langle \text{hex-digit} \rangle \mid 0 \langle \text{oct-digit} \rangle \langle \text{oct-digit} \rangle \langle \text{oct-digit} \rangle \mid \text{s} \mid \text{d} \mid \text{e}$

 $\langle \text{string-literal} \rangle ::= " \{ \langle \text{string-elem} \rangle \} "$
 $\langle \text{string-elem} \rangle ::= \langle \text{char} \rangle$ “but not an unescaped " character”

 $\langle \text{eol} \rangle ::= \langle \text{newline} \rangle \mid \langle \text{cr} \rangle \langle \text{newline} \rangle$
 $\langle \text{newline} \rangle ::= \text{“A newline character (ASCII 10)”}$
 $\langle \text{cr} \rangle ::= \text{“A carriage return character (ASCII 13)”}$

 $\langle \text{ignored} \rangle ::= \langle \text{blank} \rangle \mid \langle \text{tab} \rangle \mid \langle \text{vt} \rangle \mid \langle \text{formfeed} \rangle \mid \langle \text{comment} \rangle$
 $\langle \text{blank} \rangle ::= \text{“A space character (ASCII 32)”}$
 $\langle \text{tab} \rangle ::= \text{“A horizontal tab character (ASCII 9)”}$
 $\langle \text{vt} \rangle ::= \text{“A vertical tab character (ASCII 11)”}$
 $\langle \text{formfeed} \rangle ::= \text{“A formfeed character (ASCII 12)”}$
 $\langle \text{comment} \rangle ::= \langle \text{comment-start} \rangle \{ \langle \text{non-nl} \rangle \}$
 $\langle \text{comment-start} \rangle ::= \#$
 $\langle \text{non-nl} \rangle ::= \text{“Any character except a newline”}$

Figure 1: Lexical grammar of the SSM assembler.

```

<program> ::= <text-section> <data-section> <stack-section> .end
<text-section> ::= .text <entry-point> {<asm-instr>} <asm-instr>
<entry-point> ::= <addr>
<addr> ::= <label> | <unsigned-number>
<label> ::= <ident>
<asm-instr> ::= <label-opt> <instr> <eol>
<label-opt> ::= <label> : | <empty>
<empty> ::=
<instr> ::= <no-arg-instr> | <two-reg-comp-instr> | <no-target-offset-instr> | <no-source-offset-instr>
           | <one-reg-offset-arg-instr> | <one-reg-arg-instr> | <one-reg-offset-instr> | <shift-instr>
           | <arg-only-instr> | <immed-arith-instr> | <immed-bool-instr> | <branch-test-instr>
           | <jump-instr> | <syscall-instr>
<no-arg-instr> ::= NOP | JMP | RTN
<two-reg-comp-instr> ::= <two-reg-comp-op> <reg> , <offset> , <reg> , <offset>
<two-reg-comp-op> ::= ADD | SUB | CPW | AND | BOR | NOR | XOR | SCA | LWI | NEG
<offset> ::= <number>
<number> ::= <sign> <unsigned-number>
<sign> ::= + | - | <empty>
<no-target-offset-instr> ::= <no-target-offset-op> <reg> , <reg> , <offset>
<no-target-offset-op> ::= LWR
<no-source-offset-instr> ::= <no-source-offset-op> <reg> , <offset> , <reg>
<no-source-offset-op> ::= SWR
<one-reg-offset-arg-instr> ::= <one-reg-offset-arg-op> <reg> , <offset> , <arg>
<one-reg-offset-arg-op> ::= LIT
<arg> : <number>
<one-reg-arg-instr> ::= <one-reg-arg-op> <reg> , <arg>
<one-reg-arg-op> ::= ARI | SRI
<arg> : <number>
<one-reg-offset-instr> ::= <one-reg-offset-op> <reg> , <offset>
<one-reg-offset-op> ::= MUL | DIV | CFHI | CFLO | JMP | CSI
<shift-instr> ::= <shift-op> <reg> , <offset> , <shift>
<shift-op> ::= SLL | SRL
<shift> ::= <unsigned-number>
<arg-only-instr> ::= <arg-only-op> <arg>
<arg-only-op> ::= JREL
<immed-arith-instr> ::= <immed-arith-op> <reg> , <offset> , <immed>
<immed-arith-op> ::= ADDI
<immed> ::= <number>
<immed-bool-instr> ::= <immed-bool-op> <reg> , <offset> , <uimmed>
<immed-bool-op> ::= ANDI | BORI | NORI | XORI
<uimmed> ::= <unsigned-number>
<branch-test-instr> ::= <branch-test-op> <reg> , <offset> , <immed>
<branch-test-op> ::= BEQ | BGEZ | BGTZ | BLEZ | BLTZ | BNE
<jump-instr> ::= <jump-op> <addr>
<jump-op> ::= JMPA | CALL

```

Figure 2: The (context free) grammar of the SSM assembler, part 1 of 2.

```

<syscall-instr> ::= <offset-only-syscall> | <reg-offset-syscall> | <no-arg-syscall>
<offset-only-syscall> ::= EXIT <offset>
<reg-offset-syscall> ::= <reg-offset-syscall-op> <reg> , <offset>
<reg-offset-syscall-op> ::= PSTR | PCH | RCH
<no-arg-syscall> ::= STRA | NOTR
<data-section> ::= .data <static-start-addr> { <static-decl> }
<static-start-addr> ::= <unsigned-number>
<static-decl> ::= <data-size> <ident> <initializer-opt> <eol>
<data-size> ::= WORD | CHAR | STRING [ <unsigned-number> ]
<initializer-opt> ::= = <number> | <char-literal> | <string-literal> | <empty>
<stack-section> ::= .stack <stack-bottom-addr>
<stack-bottom-addr> ::= <unsigned-number>

```

Figure 3: The (context free) grammar of the SSM assembler, part 2 of 2.

3 Initial Values

The initial value of the program counter (*PC*) is set to the address of the program’s entry point (i.e., the value of *<entry-point>*), which is declared at the beginning of the *<text-section>*.

The start of the global data in memory is at the word address given by the data section’s static data start address (i.e., the value of *<static-start-addr>*), declared at the beginning of the *<data-section>*; this value is used as the initial value of the *\$gp* register. The data declared in the data section all have offsets from this address that are computed in declaration order, with *WORD* sized data and *CHAR* sized data both declaring a word of storage (i.e., 4 bytes) and *STRING* data taking the number of *words* declared for it.

The “bottom” of the runtime stack is given in a declaration in the stack section (*<stack-section>*); it is the value of *<stack-bottom-addr>* that follows the *.stack* keyword. This must be strictly greater than the static data start address; it is also the initial value put in the *\$fp* and *\$sp* registers at the start of a program’s execution.

4 Constraints on Assembly Code

There are some constraints on programs that the assembler checks; the assembler considers violations of these constraints to be an error.

The program’s entry point must be strictly less than the static data start address and the static data start address must be strictly less than the stack bottom address.

Furthermore, immediate operands and offsets must fit in the number of bits allowed by the instruction’s format.

References

- [1] Gerry Kane and Joe Heinrich. *MIPS RISC architectures*. Prentice-Hall, Inc., 1992.